

第4回 在庫バンク方式

高桑 宗右エ門, 三輪 冠奈

1. はじめに

本シリーズでは、在庫管理の具体的な場面を取り上げ、PCを用いて、それぞれの状況に合った在庫管理方式を適用するために、在庫管理方式およびモデル化の考え方を具体的に解説している。そして、在庫管理方式をいっそう理解しやすくするための手段を提示し、併せて、実際の問題へ応用するための参考事例を紹介することを目的とする。これまでに、第1回：定量発注方式（発注点方式）、第2回：定期発注方式、および第3回：2ピン方式（ダブル・ピン方式）を取り上げた。第4回の今回は、季節変動などにみられるように、対象期間中に需要パターンが変動するような状況に対応する一つの在庫管理方式を取り上げることにする。

本シリーズにおいて、在庫管理方式のモデル化に際しては、初めにExcelを用いて、シミュレーション・モデルを構築する。次に、グラフ上に時々刻々変化する様子を動的的に表現するために、構築済みのExcelモデルにアニメーション・モデルを融合させるための手順について解説する。本シリーズでは、Excelと併用するシステムとして、Arenaシミュレーション・アニメーション・システムを用いてモデル構築を進めることにする¹。

2. 在庫バンク方式

前回までの在庫管理方式では、対象期間中、需要は

たかくわ そうえもん, みわ かな
名古屋大学 大学院経済学研究科
〒464-8601 名古屋市千種区不老町

¹ Microsoft Excelは米国Microsoft Corporationの登録商標であり、Arenaは米国Rockwell Software Corporationの登録商標である。本シリーズで取り上げるプログラムを実行するには、これら二つのソフトウェアがインストール済みであることが必要である。Arenaソフトウェアは文献[2]の付録に添付されており、インストールに特別の制約はない。

安定していて、平均の周りに変動するものとして、各在庫管理方式について考察した。しかし、需要の基本値に増加ないしは減少傾向がみられたり、あるいは季節変動の傾向がみられたりするような場合、在庫管理を実施するには、適切に対応することが必要である。そのための一つの在庫管理方式として、在庫バンク方式（The inventory-bank system）を取り上げることにする。この方式では、調査時点で、每期、在庫補充水準を計算し、期末の手持ち在庫量と発注済量の和が在庫補充水準よりも少ないときに、その差を発注量として発注する。ここに、在庫補充水準は、直近のある期間（ M 期）の平均需要量を何倍（ N 倍）保持するかによって決定される。ここに、 M と N のパラメータの値をあらかじめ設定しておく必要がある。なお、本方式の詳細については、文献[1]を参照してほしい。

2.1 記号

本節（Excelモデル）で用いる記号を次にまとめて示す。ここでは、1期を1日とする。

a ：発注前における期末の手持ち在庫量と発注済量の和

ad ：単位期間当たりの需要量の平均

AII ：平均在庫量

ASQ ：平均品切れ量

$A1$ ：期平均在庫量

$A2$ ：期平均品切れ量

$A3$ ：期発注回数

$A4$ ：期品切れ回数

b ：在庫補充水準

ba ：在庫補充水準からの差異

C ：発注費（1回当たり）

D ：年間需要量

d ：需要量（1期）

ek ：実行期間

f ：平均需要量を算出するための変数

g ：発注済量を算出するための変数

h : 平均需要量または発注済量を計算するための期を表す変数

I : 在庫量

i : 年間在庫保管費率

$IBSDATA$: セルの名称

k : 期

ki : 通年の期

L : 納入リードタイム

M : 平均需要量を算出する期間数

$month$: 月を表す変数

N : 在庫補充水準を算出する期間数

OC : 発注回数

P : 製造原価 (1個あたり)

pr : 発注量

qb : 期首の手持ち在庫量

qe : 期末の手持ち在庫量

r : 納入量

sb : 平均需要量

SC : 品切れ回数

sd : 単位期間当たりの需要量の標準偏差

$year$: 年を表す変数

θ : 需要へのサービス率を表す信頼係数

2.2 計算式

在庫バンク方式で用いられる計算式について、計算手順に従って説明する。

(1) 期

時間単位を k 期として、ここでは1日を1期とする。

(2) 期首手持ち在庫量

前期末時点での手持ち在庫量と前期の納入量の和が当期首手持ち在庫量である。

$$qb_k = qe_{k-1} + r_{k-1} \quad (1)$$

(3) 需要量

第 k 期の需要量を d_k で示す。

(4) 期末手持ち在庫量

期首手持ち在庫量から需要を差し引いて期末在庫量が得られる。

$$qe_k = qb_k - d_k \quad (2)$$

(5) 第 k 期までの直近 M 期間の平均需要量

第 k 期を含めて、直近 M 期間の需要量の平均を計算する。

$$sb_k = \frac{1}{M} \sum_{j=k-M+1}^k d_j \quad (3)$$

(6) 在庫補充水準

在庫補充水準は、直近 M 期間の平均需要量の N 倍として設定し、次式で与えられる。

$$b_k = N \times sb_k \quad (4)$$

(7) 当該期発注前における、期末の手持ち在庫量と発注済量の和

当該期の期末手持ち在庫量と発注済 (未納入) 量の和を、次式で計算する。 L は納入リードタイムである。

$$a_k = \begin{cases} qe_k & (L=0) \\ qe_k + \sum_{j=k-L}^{k-1} pr_j & (L>0) \end{cases} \quad (5)$$

(8) 在庫補充水準からの差異

在庫補充水準から期末の手持ち在庫量と発注済 (未納入) 量の和の差異を求める。

$$ba_k = b_k - a_k \quad (6)$$

(9) 発注量

発注量は、式(6)の値と0と比較し、大きいほうを発注量とし、次式で与えられる。

$$pr_k = \max[ba_k, 0] \quad (7)$$

(10) 期末の納入量

発注済量は、納入リードタイムを経て納入される。

$$r_k = pr_{k-L} \quad (8)$$

(11) 期平均在庫量

期当たり平均在庫量および平均品切れ量は、期首および期末の手持ち在庫量により、次の諸式で計算される。なお、前稿までの各在庫管理方式においても同様の計算式が用いられる。

$$A1_k = \begin{cases} (qb_k + qe_k)/2 & (qe_k \geq 0) \\ (qb_k)^2/2d_k & (qe_k \leq 0 \text{ and } qb_k \geq 0) \\ 0 & (qe_k \leq 0 \text{ and } qb_k \leq 0) \end{cases} \quad (9)$$

(12) 期平均品切れ量

$$A2_k = \begin{cases} 0 & (qe_k \geq 0) \\ (qe_k)^2/2d_k & (qe_k \leq 0 \text{ and } qb_k \geq 0) \\ -(qb_k + qe_k)/2 & (qe_k \leq 0 \text{ and } qb_k \leq 0) \end{cases} \quad (10)$$

(13) 期発注回数

$$A3_k = \begin{cases} 0 & (a_k \geq b_k) \\ 1 & (a_k < b_k) \end{cases} \quad (11)$$

(14) 期品切れ回数

$$A4_k = \begin{cases} 0 & (qe_k \geq 0) \\ 1 & (qe_k < 0) \end{cases} \quad (12)$$

2.3 事例

ある日用品メーカーでは、42種類の製品 (洗剤、漂




図1 1日当たりの平均需要量

白剤、シャンプーなど)を生産し、販売している。今回は、原材料メーカ、メーカ(工場)・流通センタ(営業所兼務)・小売店・顧客というサプライチェーンにおいて、ある製品をとりあげて、ある流通センタにおける在庫管理について考える。

このメーカでは、5カ所の流通センタからの輸送依頼に応じて製品の輸送を行っている。メーカから、ある流通センタへの輸送にかかわる納入リードタイムは1日である。いま、製品番号6の製品について、ある流通センタにおける各月1日当たりの需要量の平均を図1に示す。図1に示すように、月ごとの需要変動がみられ、製品の需要量は各月ごとでそれぞれ異なるパラメータの正規分布に従うことがわかっている。

この製品の在庫管理において、需要変動に応じて、そのつど変動する在庫補充水準に基づいて発注量を決定する在庫バンク方式を適用し、在庫管理シミュレーションを実施することにする。

3. シミュレーション・モデル

前回までの方式を踏襲し、今回の在庫管理方式のシミュレーション・アニメーション・モデルに関する手順は、次の2ステップで構成される。

- (1) Excelによるシミュレーション実行。
- (2) ArenaによるExcelのシミュレーション結果の読みもと、在庫推移を表示するアニメーションの実行。

3.1 Excelによるシミュレーション・モデル

(1) モデルのロジックとプログラム

ロジックのフローチャートを図2に示し、VBAプログラムの詳細を図3に示す。

(2) データ入力

図4に示す画面において、シミュレーション実行に




図2 ロジックのフローチャート

必要な数値をセルに入力する。まず、セル(D6:D16)に示した各項目に対応するデータを入力する。次に、1月から12月までの各月において、需要量の平均および標準偏差をセル(G6:H17)に入力する。(3)プログラムの実行

シミュレーションを実行するために、Goボタンをクリックすると、設定画面が表示される。実行期間(図4では365)を入力し、OKボタンをクリックすると、実行が開始される。

(4) 結果表示

実行が終了すると、シミュレーション結果が表示される。プログラムの実行により、平均在庫量、平均品切れ量、発注回数などのシミュレーション実行結果に関する総括がセル(C31:C37)に表示される。さらに、在庫量の変動などの詳細の内容については、他のシート(IFS)に表示される。

3.2 シミュレーション結果

ここでは、在庫バンク方式において、各月の需要量

```

Private Sub IBS_Click()
    '変数の定義
    Dim k, ki, ek, f, g, h, I, N, M, L, d, ad, sd, OC, SC, _
        AII, ASQ, year, month As Integer

    Dim IBSDATA As String
    'ユーザーフォームの表示
    UserForm1.Show
    ek = UserForm1.Period.Value
    '初期値設定
    AII = 0
    ASQ = 0
    year = 1
    month = 1
    ad = Cells(month + 5, 7).Value
    sd = Cells(month + 5, 8).Value
    L = Cells(12, 4).Value
    I = Cells(14, 4).Value
    M = Cells(15, 4).Value
    N = Cells(16, 4).Value
    k = 1
    Worksheets("IBS").Activate
    '需要平均をとるための需要生成 (M期間)
    Do Until k = M + 1
        Randomize '乱数発生ルーチンを初期化
        d = Application.WorksheetFunction.NormInv(Rnd(), ad, sd)
        If Int(d) > 0 Then
            ActiveSheet.Cells(k + 1, 3).Value = Int(d)
        Else
            ActiveSheet.Cells(k + 1, 3).Value = 0
        End If
        ActiveSheet.Cells(k + 1, 9).Value = 0
        k = k + 1
    Loop
    ActiveSheet.Cells(k, 4).Value = I
    k = k + 1
    '実行期間終了まで各数値の計算を行う。
    Do Until k = ek + M + 2
        ActiveSheet.Cells(k, 1).Value = k - M - 1 '期(k)
        ActiveSheet.Cells(k, 2).Value = ActiveSheet.Cells(k - 1, 4).Value + _
            ActiveSheet.Cells(k - 1, 10).Value '期前の手持ち在庫量(z(k))
        '需要生成
        Randomize '乱数発生ルーチンを初期化
        d = Application.WorksheetFunction.NormInv(Rnd(), ad, sd)
        If Int(d) > 0 Then
            ActiveSheet.Cells(k, 3).Value = Int(d) '需要(d(k))
        Else
            ActiveSheet.Cells(k, 3).Value = 0
        End If
        ActiveSheet.Cells(k, 4).Value = ActiveSheet.Cells(k, 2).Value - _
            ActiveSheet.Cells(k, 3).Value '期末の手持ち在庫量(qe(k))

        '直前M期間の需要合計の計算
        h = k
        Do Until h = k - M
            f = f + ActiveSheet.Cells(h, 3).Value
            h = h - 1
        Loop

        ActiveSheet.Cells(k, 5).Value = Int(f / M) '直前M期の平均需要(sb(k))
        f = 0
        ActiveSheet.Cells(k, 6).Value = N * ActiveSheet.Cells(k, 5).Value
            '在庫補充水準(b(k))

        '発注量の計算
        h = k
        g = 0
        Do Until h = k - L
            g = g + ActiveSheet.Cells(h - 1, 9).Value
            h = h - 1
        Loop

        ActiveSheet.Cells(k, 7).Value = ActiveSheet.Cells(k, 4).Value + g
            '発注前の期末の手持ち在庫量と発注量の和(a(k))
        g = 0
        ActiveSheet.Cells(k, 8).Value = ActiveSheet.Cells(k, 6).Value - _
            ActiveSheet.Cells(k, 7).Value '在庫補充水準からの差異(ba(k))

        If ActiveSheet.Cells(k, 8).Value > 0 Then '差異が0より大きいとき
            ActiveSheet.Cells(k, 9).Value = ActiveSheet.Cells(k, 8).Value
            '発注量(pr(k)=ba(k))
        Else
            ActiveSheet.Cells(k, 9).Value = 0 '発注量は0(pr(k)=0)
        End If

        ActiveSheet.Cells(k, 10).Value = ActiveSheet.Cells(k - 1, 9).Value
            '期末の在庫量(r(k))

        '期平均在庫量の計算(A1)
        If ActiveSheet.Cells(k, 4).Value > 0 Then
            ActiveSheet.Cells(k, 11).Value = (ActiveSheet.Cells(k, 2).Value + _
                ActiveSheet.Cells(k, 4).Value) / 2
        Else
            If ActiveSheet.Cells(k, 2).Value > 0 Then
                ActiveSheet.Cells(k, 11).Value = (ActiveSheet.Cells(k, 2).Value * _
                    ActiveSheet.Cells(k, 2).Value) / (2 * ActiveSheet.Cells(k, 3))
            Else
                ActiveSheet.Cells(k, 11).Value = 0
            End If
        End If

        AII = AII + ActiveSheet.Cells(k, 11).Value

        '期平均品切れ量の計算(A2)
        If ActiveSheet.Cells(k, 4).Value > 0 Then
            ActiveSheet.Cells(k, 12).Value = 0
        Else
            If ActiveSheet.Cells(k, 2).Value > 0 Then
                ActiveSheet.Cells(k, 12).Value = (ActiveSheet.Cells(k, 4).Value * _
                    ActiveSheet.Cells(k, 4).Value) / (2 * ActiveSheet.Cells(k, 3))
            Else
                ActiveSheet.Cells(k, 12).Value = -(ActiveSheet.Cells(k, 2).Value + _
                    ActiveSheet.Cells(k, 4).Value) / 2
            End If
        End If

        ASQ = ASQ + ActiveSheet.Cells(k, 12).Value

        '期発注回数計算(A3)
        If ActiveSheet.Cells(k, 7).Value > ActiveSheet.Cells(k, 6).Value Then
            ActiveSheet.Cells(k, 13).Value = 0
        Else
            ActiveSheet.Cells(k, 13).Value = 1
            OC = OC + 1
        End If

        '期品切れ回数計算(A4)
        If ActiveSheet.Cells(k, 4).Value > 0 Then
            ActiveSheet.Cells(k, 14).Value = 0
        Else
            ActiveSheet.Cells(k, 14).Value = 1
            SC = SC + 1
        End If

        '月ごとにadとsdを変更
        ki = (k - M - 1) - 365 * (year - 1)
        Select Case ki
            Case 31, 59, 90, 120, 151, 181, 212, 243, 273, 304, 334
                month = month + 1
                ad = Cells(month + 5, 7).Value
                sd = Cells(month + 5, 8).Value
            Case 365
                year = year + 1
                month = 1
                ad = Cells(month + 5, 7).Value
                sd = Cells(month + 5, 8).Value
        End Select

        Worksheets("IBS").Activate
        k = k + 1
    Loop

    '最初に生成した需要の削除
    k = 1
    Do Until k = M + 1
        ActiveSheet.Range("A2:n2").Delete (xlShiftUp)
        k = k + 1
    Loop

    '結果記入
    Cells(31, 3).Value = AII / ek
    Cells(32, 3).Value = ASQ / ek
    Cells(33, 3).Value = OC
    Cells(34, 3).Value = SC
    'セルの名称定義
    Worksheets("IBS").Range(Worksheets("IBS").Range("A4"), _
        Worksheets("IBS").Range("A4").End(xlToRight)). _
        End(xlDown).Name = "IBSDATA"
    Worksheets("Sheet1").Select
End Sub

```

図3 VBAプログラムの詳細

の平均と標準偏差から需要量を生成し、納入リードタイム(L)は1日、平均需要量を算出する期間(M)は5日、在庫補充水準を算出する期間(N)は3日分として、365日についてシミュレーションを実行した。図5は、シート (IBS) に書き出された結果である。図

中、第10期(日)について、計算過程を追ってみる。まず、期首の手持ち在庫量は321(以下、単位は個)で、その日の需要量は104であるので、期末在庫量は217となる。第6ないし10期までの平均需要量は151であるので、在庫補充水準は453となる。一方、前日(第9期)に発注してこの日に納入する数量は144であるので、期末在庫量と合わせて361となる。したがって、92(=453-361)が当期の発注量となる。なお、その期の平均在庫量は、期首および期末の在庫量の平均として計算され、269である。

The Inventory-Bank System
在庫バンク方式

Go Clear

データの入力

パラメータ(Parameter)	記号(Sign)	数値(Value)
製造原価(1個当り)	P	300
年間在庫保管費率	i	10.00%
発注費用(1回当り)	C	500
年間発注量	D	4922.40
期当りの需要量の標準偏差	σd	27.03
期当りの需要量の平均	μd	135.13
納入リードタイム	L	1
サービス率を表す信頼係数	β	1.65
初期在庫量	I	200
平均需要量を算出する期間数	M	5
在庫補充水準を算出する期間数	N	3

1日当りの需要量

月	平均	標準偏差
1	152.8	61.10
2	111.6	44.62
3	164.1	65.65
4	160.5	64.21
5	118.5	47.39
6	110.1	44.05
7	122.1	49.84
8	122.8	49.12
9	115.7	46.29
10	128.6	51.44
11	129.7	51.87
12	182.1	72.83

設定

実行期間: 365

シミュレーション結果

	数値(Value)
平均在庫量	
平均品切れ量	
発注回数	
品切れ回数	
発注費用	0
保管費用	0
総変動費用	0

図4 Excelによるシミュレーション設定画面

4. アニメーション・モデル

4.1 記号

本節(Arenaモデル)で用いる記号を次にまとめて示す。なお、節2.1で既出の記号については省略する。Arenaでは、Excelモデルで使用した添え字と記号末尾のkは対応している。

ak: 発注前における期末の手持ち在庫量と発注量の和

bak: 在庫補充水準からの差異

bk: 在庫補充水準(kk期)

dk: 需要量(kk期)

	A	B	C	D	E	F	G	H	I	J	K	L	M	N
期	期首手持ち在庫量	需要	期末手持ち在庫量	直前M期の平均需要	在庫補充水準	発注前の期末手持ち在庫量と発注量の和	在庫補充水準からの差異	期末の発注量	期末の在庫量	期平均在庫量	期平均品切れ量	発注回数	品切れ回数	
1														
2	1	200	173	27	175	525	27	498	498	0	113.5	0	1	0
3	2	27	75	-48	144	432	450	-18	0	498	154	0	1	0
4	3	450	189	261	145	435	261	174	174	0	355.5	0	1	0
5	4	261	167	94	166	498	268	230	230	174	177.5	0	1	0
6	5	268	122	146	145	435	376	59	59	230	207	0	1	0
7	6	376	190	186	148	444	245	199	199	59	281	0	1	0
8	7	245	141	104	161	483	303	180	180	199	174.5	0	1	0
9	8	303	171	132	158	474	312	162	162	180	217.5	0	1	0
10	9	312	153	159	155	465	321	144	144	162	235.5	0	1	0
11	10	321	104	217	151	453	361	92	92	144	269	0	1	0
12	11	361	92	269	132	396	361	35	35	92	315	0	1	0
13	12	361	195	166	143	429	201	228	228	35	263.5	0	1	0
14	13	201	196	5	148	444	233	211	211	228	103	0	1	0
15	14	233	251	-18	167	501	193	308	308	211	108.15	0.65	1	1
16	15	193	151	42	177	531	350	181	181	308	117.5	0	1	0
17	16	350	224	126	203	609	307	302	302	181	238	0	1	0
18	17	307	134	173	191	573	475	98	98	302	240	0	1	0
19	18	475	205	270	193	579	368	211	211	98	372.5	0	1	0
20	19	368	127	241	168	504	452	52	52	211	304.5	0	1	0
21	20	452	92	360	156	468	412	56	56	52	406	0	1	0
22	21	412	325	87	176	528	143	365	365	56	249.5	0	1	0
23	22	143	94	49	168	504	434	70	70	365	96	0	1	0
24	23	434	122	312	152	456	382	74	74	70	373	0	1	0
25	24	382	145	237	155	465	311	154	154	74	309.5	0	1	0
26	25	311	163	148	169	507	302	205	205	154	229.5	0	1	0
27	26	302	260	42	156	468	247	221	221	205	172	0	1	0

図5 Excelによるシミュレーション結果




図6 アニメーション・モデルの詳細

kk: 期²

line: 0 を表示するためのプロット変数

prk: 発注量

qbk: 期首の手持ち在庫量

qek: 期末の手持ち在庫量

qq: 需要量のプロット変数

rk: 納入量

sbk: 第 kk 期までの直近 M 期間の平均需要量

4.2 モデルの実行手続き

次の手順で、アニメーションを実行する。

[Step 1] エンティティを生成する。1 期 (日) に 1 個のエンティティが生成される。

[Step 2] 実行された Excel ファイルを読み込む。

[Step 3] プロットに表示する在庫量 (qq) と 0 を表示するプロット変数 (line) を割り付ける。

[Step 4] 発注回数と品切れ回数をカウントする。

[Step 5] エンティティを破棄する。

² Arena では、期を表す変数は kk を用いる。これは、節 2.1 の k に対応している。




図7 Plot のダイアログ

モデルの詳細を図 6 にまとめて示す。図中、上側のブロック線図はプログラムのモジュール表示である。

4.3 アニメーション図の詳細

前回までと同様に、アニメーションにプロットと変

数 (Variable) を表示する。これまでのモデルでは、Plot ダイアログ内の Expression において、二つの変数 (qq , $line$) を設定した。今回のモデルでは、在庫補充水準もプロット表示を行うため、図7に示すように、変数 bk を追加する。また、変数 (Variable) の表示についても、変数 bk を Variable ボタンを用いて追加する。

4.4 アニメーションの実行

Excel シミュレーション・モデル実行後に、アニメーション・モデルにより、シミュレーション結果をダイナミックに表示する。実行の様子を図8に示す。図中、上側に表示されているプロットは、在庫補充水準

の推移の様子であり、下側に表示されているプロットは手持ち在庫量の推移の様子を表している。

4.5 在庫管理方式の比較

これまでに取り上げた四つの在庫管理方式について、表1にまとめて示す。表中、各方式の特徴として、発注量、発注間隔、在庫補充水準についての相違を示し、本シリーズで特に詳しく述べた Arena アニメーションに関する諸設定についても併せて示す。さらに、各モデルで用いたパラメータについても、表2にまとめて示す。




図8 在庫バンク方式のシミュレーション・アニメーション・モデルの実行の様子

表1 在庫管理方式の比較

項目 在庫管理方式	管理方式			詳述したアニメーション機能
	発注量	発注間隔	在庫補充水準	
(1) 定量発注方式	一定	不定期	—	Plot機能
(2) 定期発注方式	変動	定期	一定	Variable機能
(3) ダブルピン方式	一定	不定期	一定	Level機能
(4) 在庫バンク方式	変動	不定期	変動	Plot機能

表2 パラメータの一覧

	Excelモデルのパラメータ			Arenaモデルのパラメータ			
(1) 定量発注方式	ad, All, ASQ, d, ek, l, k, L, OC, SC, sd	C, D, i, P, θ , SA	EQQ, FOASDATA, ss	A1, A2, A3, A4, dk, line, prk, qbk, qek, qq, rk	qrk	kk	
(2) 定期発注方式			POSDATA, ST, T				
(3) ダブルピン方式		QR	BINDATA, BinNo, dq, l1, l2, lq, m, n, nn, TO				bin1, bin2, Minutes, Seconds
(4) 在庫バンク方式		C, D, i, P, θ	f, g, h, ki, M, month, N, IBSDATA, sb, year				kk, ak, bak, bk, sbk

5. おわりに

本シリーズでは、在庫管理方式の基本モデルについて、事例を取り上げて、シミュレーション・モデルを構築する方法について述べた。たしかにExcelを用いたシミュレーションでは、数値実験だけでなく、グラフを描くことも可能であるが、グラフが一度にすべて表示されてしまうという難点がある。そこで、特にその難点を克服するために、時間進行とともに刻々と在庫量などの変動の様子を表示するためのモデル化の考え方を提示した。したがって、在庫管理だけでなく、オペレーションズ・リサーチの種々のテーマにも応用が可能であるので、参考となれば欣幸である。

プログラムおよび実行の様子を下記 URL に掲示す

るので、本稿に関する内容の詳細については参考なし照会してほしい。

<http://www.stlab.soec.nagoya-u.ac.jp/>

E-mail: takakuwa@soec.nagoya-u.ac.jp

参考文献

- [1] Eliezer Naddor: *Inventory Systems*, John Wiley & Sons, 1966.
- [2] W. D. Kelton, R. P. Sadowski, and D. A. Sadowski: *Simulation with Arena*, 2nd ed., McGraw-Hill, 2001 (高桑宗右エ門監訳: “シミュレーション—Arenaを活用した総合的アプローチ” (第2版), コロナ社, 2002).